

IDENTIFICATION

Product Code: MAINDEC-08-D2AA-D

Product Name: PDP-8 Teletype Reader Test

Date Created: April 12, 1965

Maintainer: Diagnostic Group

Author: M. Horovitz

Previous Code: MAINDEC 810

0.

0

0

1. ABSTRACT

MAINDEC-08-D2AA tests the performance of the Teletype Model 33 Perforated Tape Reader, using the reader to scan a closed-loop test tape punched with alternating groups of the character codes 000 and 377.

Each character is tested for bits either dropped or gained while reading; each group of characters is checked for characters missed entirely or read more than once.

An Auxiliary Punch program (see paragraphs 3.3.1 and 5.1.2) is available to produce the tape used in the test.

2. REQUIREMENTS

Storage

MAINDEC-08-D2AA requires the RIM Loader to be in memory locations 7756-7776 octal. The test program occupies memory locations 0001-0122 octal.

Equipment

Teletype Model 33 Keyboard/Printer with tape punch and reader.

3. USAGE

3.1 Loading

RIM Loader must be in memory.

With the Tape Reader off, set the SWITCH REGISTER to 7756 octal. Press LOAD ADDRESS, then START.

Place the MAINDEC-08-D2AA Test Tape in the Teletype reader, and start the reader.

When loading is completed, turn off the reader and remove the program tape.

3.2 Switch Settings

Starting Addresses (in octal):

0100 Auxiliary Punch program

0001 Reader Test program

3.3 Start-up and/or Entry

3.3.1 Auxiliary Punch Program - The Teletype reader must be off. Set the SWITCH REGISTER to 0100 octals. Press LOAD ADDRESS, then START. Turn on the Teletype punch.

Allow the program to punch several feet of tape, then stop the computer, turn off the punch, and tear off the tape.

3.3.2 Reader Test Program - Place the leading end of the test tape, (produced by the Auxiliary Punch program) in the Teletype reader. Set the SWITCH REGISTER to 0001. Press LOAD ADDRESS, then START. Turn on the reader.

3.4 Errors in Usage

3.4.1 Error Stops

Error	C(MA)	Cause of Error
1	0025	The last character read was incorrect. The AC contains the erroneous code. The correct character is found on the test tape in the <u>second</u> one past the read station. If this is 337, the AC shows dropped bits as 0's. If the correct code is 000, the AC shows spurious bits as 1's.
2	0055	Character count is incorrect. The program has just finished reading a group of 337 codes; the number of characters in this group is not identical to the number of characters in the preceding group of 000 codes. The number in the AC is the difference between the counts. The error was probably the result of a missed character or a failure in the transmission of a pulse from the reader buffer to the AC.

3.5 Recovery from Such Errors

Error	Recovery
1	To restart the test, press CONTINUE.
2	To restart the test, press CONTINUE.

4. RESTRICTIONS

None. The test will operate on a PDP-8 of minimum configuration.

5. DESCRIPTION

5.1 Discussion

5.1.1 The Reader Test Program - MAINDEC-08-D2AA tests the perforated tape reader associated with the Teletype Model 33 Keyboard/Printer. The program causes the reader to continuously scan a specially prepared test tape and checks for the accurate transmission of each character from the read heads into the AC. A dropped bit (0 transmitted instead of a 1) or a spurious bit (1 transmitted instead of a 0) will cause the program to halt. The erroneous character will appear in the AC.

The loop contains a sequence of character codes, each of which is either 000 (no holes punched) or 377 (all holes punched). The characters are arranged as follows: A group of 000 characters is followed by a group containing an identical number of 377's. The size of each successive pair of groups may vary. A sample sequence is shown in section 5.1.2.

The test program counts the number of characters in a group of 000's and then decreases this count while reading the subsequent group of 377's. At the end of that group, the count should reduce to 0. If it does not, the test halts; the probable cause of error is that the reader missed a character entirely. If the count is still positive, the missed code was 377; if negative, it was 000.

In either case, the test will be restarted when the operator presses CONTINUE.

5.1.2 Auxiliary Punch Program - This is a separate program within MAINDEC-08-D2AA which punches a length of tape for use in the main test program. This tape contains the following pattern, repeated as many times as desired:

Tape Pattern	Octal Equivalent
00000 000	000
11111 111	377
00000 000	000
00000 000	000
11111 111	377
11111 111	377
00000 000	000
00000 000	000
00000 000	000
11111 111	377
11111 111	377
11111 111	377

There is no halt in the punch program. Once started, it will run continuously until the operator stops it, allowing him to produce as long a test tape as he wishes.

5.2 Examples and/or Applications

Although the primary use of MAINDEC-08-D2AA is to test the performance of the tape read heads and the transmission lines between the read heads, reader buffer, and AC, it can also be used as a quick check of the tape advance mechanism. If, after a few feet of tape have been read, there is evidence of wear or of enlarged feed holes, there may be a mechanical malfunction. Normally the test would be used as a preventive maintenance procedure or when program operation tends to indicate reader malfunction.

6. METHODS

At the start of the test, the program insures that the tape is positioned at the beginning of a group of 000 characters; it examines the tape until a nonzero character appears, then continues to read until the first 000 character is read. This character is tested for spurious bits and the counter is incremented. The next character is read and tested. It is then compared with the previous character to determine if the two are identical. If so, the count is incremented, and the next character is read and tested. This procedure continues until a character is read which differs from the previous one. At this point, two possibilities become apparent:

If the previous character was 000 and the current one is 377, the character count is decremented, and the next character is read and tested. The test proceeds until the next change of character.

If the previous character was 377 and the current one is 000, the count is checked for missed characters. If the count is 0, there has been no error and the test continues. If the count is not 0, the program halts with the value of the count in the AC. If a line was missed while reading 000's, there will be an excess of 377's, and the count will be negative. If a line was missed while reading 377's, there will be a deficiency of these, and the count will be positive. In each case, the absolute value of the count corresponds to the number of lines which were missed.

Example: While reading a group containing three 000 characters, the reader failed to transmit the third one, so that the count remained at 2. The subsequent group of three 377's was read correctly, so that when the last of these was read, the count became -1. The program stops with 7777 (= -1) displayed in the AC lights.

7. PROGRAM LISTING

/MAINDEC-08-D2AA: PDP-8 TELETYPE READER TEST

*1

0001	4057	A,	JMS READ	/TAPE PHASING SEQUENCE
0002	7450		SNA	/IS CHARACTER NON-ZERO?
0003	5001		JMP .-2	/NO-KEEP LOOKING
0004	4057		JMS READ	/YES-LOOK FOR FIRST ZERO
0005	7440		SZA	/CHARACTER
0006	5004		JMP .-2	

0007	3072	B,	DCA TEM	/START WITH ZERO CHARACTER
0010	3074		DCA LINCT	/INITIALIZE LINE COUNTER
0011	3073		DCA PREV	/HOLDS PREVIOUS LINE
0012	5016		JMP D+3	
0013	2074	D,	ISZ LINCT	/INCREMENT COUNTER WHEN C(TEM)=0
0014	4057		JMS READ	
0015	3072		DCA TEM	/STORE CHARACTER JUST READ
0016	1067		TAD CT1	/TEST FOR NON-ZERO CHARACTER
0017	1072		TAD TEM	/RESULT SHOULD BE 0 IF C(TEM)=377
0020	7650		SNA CLA	/IS IT?
0021	5027		JMP OK	/YES - GO ON
0022	1072		TAD TEM	/NO - CHECK FOR ZERO
0023	7450		SNA	
0024	5027		JMP OK	/IT IS - GO ON
0025	7402	ERR1,	HLT	/ERROR: AC CONTAINS ERRONEOUS C.
0026	5001		JMP A	/TRY AGAIN
0027	1072	OK,	TAD TEM	/PREPARE FOR C(PREV)/C(TEM) COMP.
0030	7640		SZA CLA	/IS IT ZERO CHAR.?
0031	1070		TAD CT1+1	/NO - SET NEXT SKIP TO SNA
0032	1071		TAD CT1+2	/IF ZERO, NEXT SKIP IS SZA
0033	3035		DCA .+2	
0034	1073		TAD PREV	
0035	0000		0	/TEST FOR STATE OF PREVIOUS CHAR.
0036	5047		JMP COM	/EXAMINE LINE COUNT
0037	1072	SAME,	TAD TEM	/GET HERE ONLY IF C(PREV)=C(TEM)
0040	7650		SNA CLA	
0041	5013		JMP D	/IF C(TEM)=0, GO ON
0042	1074		TAD LINCT	/IF C(TEM)=377, DECREMENT COUNT
0043	7041		CIA	
0044	7040		CMA	
0045	3074		DCA LINCT	
0046	5014		JMP D+1	/NEXT CHARACTER
0047	1072	COM,	TAD TEM	/GET HERE ONLY IF C(PREV)IS NOT=
0050	7640		SZA CLA	/IS C(TEM)=0?
0051	5042		JMP COM-5	/NO-GO DECREMENT COUNT AND CONT.
0052	1074		TAD LINCT	/YES - CHECK COUNTER
0053	7450		SNA	
0054	5013		JMP D	/C(LINCT)=0, CONTINUE
0055	7402	ERR2,	HLT	/ERROR: LINE COUNT INCORRECT.
0056	5001		JMP A	/AC CONTAINS C(LINCT), TRY AGAIN

```

0057 0000 READ, 0 /CHARACTER READING SUBROUTINE
0060 7200 CLA
0061 1072 TAD TEM
0062 3073 DCA PREV /SAVE PREVIOUS CHARACTER
0063 6031 KSF /WAIT LOOP
0064 5063 JMP .-1
0065 6036 KRB
0066 5457 JMP I READ

0067 7401 CT1, 7401 /2'S COMPLEMENT OF 377 FOR CHAR.
0070 0010 I0 /SNA BIT FOR COMPARISON TEST
0071 7640 SZA CLA /COMPARISON TEST INSTRUCTION

0072 0000 TEM, 0 /HOLDS CURRENT CHARACTER
0073 0000 PREV, 0 /HOLDS PREVIOUS CHARACTER
0074 0000 LINCT, 0 /CHARACTER COUNT

*100 /AUXILIARY PUNCH PROGRAM

PAT=TEM

0100 7300 AA, CLL CLA
0101 1121 TAD CT2 /LINE COUNT FOR TAPE PATTERN
0102 3074 DCA LINCT
0103 1122 TAD CT2+1 /PATTERN SELECTOR
0104 3072 DCA PAT
0105 1072 BB, TAD PAT
0106 7004 RAL
0107 3072 DCA TEM
0110 7430 SZL
0111 7040 CMA /IF C(L)=1, PUNCH 377 CHARACTER
0112 6046 TLS /IF C(L)=0, PUNCH 0 CHARACTER
0113 6041 TSF
0114 5113 JMP .-1
0115 7200 CLA
0116 2074 ISZ LINCT /COUNT CHARACTER
0117 5105 JMP BB /NOT FINISHED
0120 5100 JMP AA /FINISHED - REPEAT PATTERN

0121 7764 CT2, 7764 /CHARACTER COUNT
0122 2307 2307 /PATTERN SELECTOR

```

```


A 0001
AA 0100
B 0007
BB 0105
COM 0047
CT1 0067
CT2 0121
D 0013
ERR1 0025
ERR2 0055
LINCT 0074
OK 0027
PAT 0072
PREV 0073
READ 0057
SAME 0037
TEM 0072

```



8. DIAGRAMS

8.1 Flow Charts

8.1.1 Character Read Routine



8.1.2 Auxiliary Punch Program



8.1.3 Reader Test Program

